
Abstract. The focus of this study is to determine what information can be obtained about
a convex body in the plane given two point X-ray functions for the body; in particular,
whether a convex body is uniquely determined by two such X-rays. We shall specifically
investigate the case in which one X-ray is taken from a source exterior to the body and the
other is taken from an interior source. We shall derive formulae for the tangent lines and
curvature of the body’s boundary at its x-intercepts, and we shall use the Stable Manifold
Theorem to design an algorithm that attempts to construct bodies with identical X-ray
functions at both sources. Finally, we shall consider the possibility of forming an analytic
argument proving the existence of exactly two bodies whose X-rays at each of the two points
are indeed equal.

1. Introduction

Geometric tomography is the branch of mathematics concerned with the reconstruction of
geometric objects from the knowledge of their sections or projections. Following the devel-
opment of medical radiology, it became natural to look at mathematical objects modeling
X-rays to study their properties, particularly whether they allow for accurate reconstructions
of geometric objects like those that might be found in the human body.

We utilize a mathematical abstraction of X-rays in this study; given a point P and a
convex body K in R

2, we consider the lengths of the intersections of K with every line
passing through P . This differs from the model of so-called directed X-rays in that we do
not know on which “side” of P the object lies (see Figure 1).

Gardner [5] lists two cases of point X-rays of a convex body that are open problems of
mathematical interest. There has already been research conducted on the first: X-rays from
two exterior sources whose connecting line intersects the body [8]. The second—X-rays from
one exterior source and one interior source—will be the topic of this paper. We will be
adapting many of the methods used by researchers of other problems of X-ray tomography
for our use in this particular case.

Figure 1. The point X-ray functions from source P for the bodies K and S
are identical, but the directed X-ray function from P for S is offset by π from
that of K.

In particular, we will attempt to determine if there is more than one body which generates
the same point X-ray data from two sources. That is, given a convex body, K, and its X-
rays, XP and XQ, from the points P and Q, we seek to find (or prove the nonexistence of) a
second convex body, S, such that the X-rays of S from P and Q are precisely XP and XQ.
We will attack this problem by first finding the x-intercepts, or “basepoints,” of S (should it
exist), approximating S by its tangent lines, and then using the Stable Manifold Theorem to
write an algorithm that will construct an approximation of S. It is our hope that, through
numerical computations, we will gather evidence that the existence of S is plausible as well
as find directions in which to pursue an analytic proof of its existence.

1.1. Definitions. We shall now introduce some definitions that will be important for later
discussion.

Definition 1.1. A convex body, K, is a compact, convex subset of R
2 with nonempty

interior; we denote its boundary ∂K.

Definition 1.2. Given a point, P ∈ R
2, and a convex body, K, the point X-ray function,

or simply X-ray, of K from P is a function, XP (ϕ), ϕ ∈ [−π
2
, π

2
), such that XP (ϕ) is the

length of ℓ ∩K, where ℓ is the line passing through P at angle ϕ.

Unless otherwise noted, all angles mentioned in this paper will be oriented counterclockwise
from the positive x-axis.

Definition 1.3. For a point, P , exterior to a convex body, K, the nearside point at angle
ϕ is the point, rP,ϕ ∈ ℓ ∩ K, closest to P , where ℓ is again the line passing through P at
angle ϕ. Similarly, the farside point at angle ϕ is the point, RP,ϕ ∈ ℓ ∩K, farthest from
P .

We also define functions of ϕ called the nearside and farside functions from P . The
nearside function, rP (ϕ), is the distance from P to rP,ϕ, and the farside function, RP (ϕ),
is the distance from P to RP,ϕ. Thus, the X-ray function from P satisfies the identity
XP (ϕ) = RP (ϕ) − rP (ϕ).

Definition 1.4. For a point Q in the interior of K, we dispense with the notions of “near-
side” and “farside” and instead define RQ(ϕ), with ϕ ∈ [−π, π), as the distance from Q
to the point farthest from Q in ρ ∩ K, where ρ is the ray originating from Q at angle of
inclination ϕ. In this case, we see that the point X-ray function from Q satisfies the identity
XQ(ϕ) = RQ(ϕ) +RQ(ϕ+ π), with ϕ in this case within [−π

2
, π

2
).

A short remark on our choice of notation: despite abandoning the terminology “farside,”
we use a capitalized ‘R’ for the function RQ. We do this to maintain consistency with RP ;
notice that RP and RQ are always concave towards their sources, while rP is concave away
from its source.

Definition 1.5. The basepoints of a convex body, K, given the points P /∈ K and Q ∈
int K, are the points of intersection in ℓ∩ ∂K, where ℓ is the line passing through P and Q.

The distances from P to the basepoints are rP (0) and RP (0), and the distances to the
basepoints from Q are RQ(0) and RQ(π). From a result of Falconer [4], we are able to
establish the locations of the basepoints given P , Q, and K; thus, we are able to determine
the values of these four distances.

Definition 1.6. Given a C2 function f in polar coordinates, the signed curvature of f ,
denoted κf , is given by the equation

κf =
f 2 + 2(f ′)2 − ff ′′

(f 2 + (f ′)2)
3

2

.

κf will be positive at θ when the graph of f is concave towards the origin at (θ, f(θ)), and it
will be likewise negative when the graph of f is concave away from the origin.

Note that curvature is independent of a curve’s parametrization; thus, changing the lo-
cation of the origin in measuring a curve by polar coordinates will not change the signed
curvature at any point (except perhaps by a factor of −1 to account for changes in concav-
ity). We will use this property when measuring the curvature of a convex body from our
two different point sources.

Further, to simplify computations of curvature, we introduce the following:

Definition 1.7. The curvature operator of f , denoted Kf , is equal to κf · (f 2 + (f ′)2)
3

2 .

Incidentally, K(−f) = Kf . This fact comes in handy for curvature calculations appearing
later in this paper.

Definition 1.8. If K is a convex body and P a point, consider the function XP (ϕ). We call
the angles α = sup{ϕ | XP (ϕ) > 0} and β = inf{ϕ | XP (ϕ) > 0} the support angles of
XP . Moreover, if ℓ is a line passing through P at angle of inclination α or β, we call ℓ a
support line of K through P .

Finally, we define a term that will be important in our investigation of a convex body’s
uniqueness with respect to its X-rays.

Definition 1.9. Given a convex body, K, and points P,Q ∈ R
2, a shadow body, S, of K

is a convex body such that its point X-ray functions from P and Q match the point X-ray
functions of K, respectively, from P and Q.

Other literature in the field may call any compact planar region with X-rays equal to those
of K a shadow body; however, we shall restrict our usage of the term only to convex objects
exhibiting this property.

1.2. An Example Convex Body. Throughout this paper we shall work with a particular
body chosen for its possession of certain “nice” properties. However, when possible, we shall
try to prove properties of convex bodies in general. The example convex body is defined as
the intersection of the two closed disks (x − 8)2 + y2 ≤ 9 and (x − 12)2 + y2 ≤ 9, with the
point P located at the origin and the point Q located at (10, 0), the centroid of the convex
body. Notice that the line passing through the two vertices of our body also contains Q.
Our body’s X-ray at angle 0 from both P and Q will be 2, and we can also calculate the

support angles of our body with respect to P : they are α = ± arccos
(
2
√

5
21

)
.

Figure 2. Our example convex body.

2. Falconer’s Lemma and Computing the Basepoints of a Shadow Body

The ultimate goal of our study is to determine whether or not two convex bodies can exist
which each have the same two X-ray functions from point sources. In other words, we may
ask whether, given a convex body and its X-rays from the points P and Q, there exists a
second convex body whose X-rays from P and Q are precisely the same as those of K. We
begin by determining where, if it does exist, such a shadow body must be located relative
to the original body. Indeed, it is possible to find the basepoints of the potential shadow
body using Lemmas 3 and 4 from Falconer [4]. By the uniqueness theorems of Falconer [4]
and Gardner [6], we know that there can be at most one convex body passing through these
particular points on the x-axis which also satisfies the X-ray functions XP and XQ. Thus,
finding these points will aid us greatly in our attempt to discover a shadow body.

Let K be a convex body in the plane, and let P and Q be points in the plane such that
the line passing through P and Q intersects the interior of K. Orient the plane so that P
lies on the origin and so that Q lies to the right of P on the x-axis. Denote by N and F the
points on ∂K that intersect the x-axis, with N nearer to P than F .

Denote by p1, q1, p2, and q2 the signed distances from P to F , from P to N , from Q to F ,
and from Q to N , respectively (see Figure 3). Let A denote the distance from P to Q (so A
is effectively p1 − p2). Also let f(t) = t log |t| − (t−m) log |t−m|, where m is the distance
from N to F (note that m = p1 − q1 = p2 − q2). Falconer’s Lemma tells us that

lim
ε→0

1

2

[∫ ε

π−ε

XQ(ϕ)

sin(ϕ)
dϕ−

∫ ε

π−ε

XP (ϕ)

sin(ϕ)
dϕ

]
= p1 log |p1| − q1 log |q1| − (p2 log |p2| − q2 log |q2|),

which happens to equal f(p1) − f(p2). We label this quantity B.

Figure 3. An example of a convex body with labeled distances between
points. Note that the distance from Q to N will have the opposite sign of
the other distances.

Now, we wish to know what other basepoints (that is, other possible locations for N and
F ) will give us the same value of B for fixed sources. Thus, following our example, we fix
A = 10. We may also fix m = 2 since we know the length of the X-ray along the x-axis
must be the same for any potential shadow body. Our question therefore reduces to finding
solutions to the equation f(p1) − f(p1 − 10) = B, where B ≈ 6.60183 in our example.

Using the FindRoot command of Mathematica 6.0, we are able to find that the solutions
of this equation are p1 = 11 (as expected) and p1 ≈ 11.1974 (see Figures 4 and 5). Thus,
the farside basepoint of our possible shadow body is approximately (11.1974, 0), and the
corresponding nearside basepoint is (9.1974, 0).

It is also worth noting that, if the original body is C1 except at two points (we shall call
them “vertices”) and the line connecting those vertices contains the point Q, the shadow
body will have vertices at exactly the same points as those of the original body. This is true
because at each vertex the X-ray functions will have discontinuous derivatives. Since the
X-rays of the shadow body are the same as those of the original body, each vertex of the
shadow body must therefore lie on the line connecting the two vertices of the original body



86 Chris Pryby

Figure 4. A plot of the Falconer function, g(t) = f(t) − f(t−A), against y = B.

Figure 5. A second plot of the above, scaled to clearly show the intersection
of g(t) with B.

as well as a line passing through P and a vertex of the original body. The intersections of
these lines are, in fact, the vertices of the original body.

3. Computing Tangent Lines at Basepoints

We will now compute the tangent lines at the basepoints of our convex body, K, using
two X-rays: one from P in the exterior of K and one from Q in the interior. This work
draws heavily from the processes used in §2 of [9] and §4 of [8], albeit with some variations
to account for our use of an interior source instead of two exterior sources.

Though in the particular problem we will be considering it is unnecessary to explicitly
recalculate the tangent lines and curvature of our convex body (as this body is given to us),
we can use this same process to calculate the tangent lines of a shadow body (presuming
one exists) at its basepoints. In fact, this is our motivation for calculating the tangent line
formulae—it will give us an initial approximation of the shadow body which we will use to
construct it.

Theorem. Let K be a convex body (C1 near its basepoints), and let P /∈ K, Q ∈ int K such
that P and Q lie on the x-axis with P to the left of Q. Then the angles of inclination of the
tangent lines at K’s basepoints are computable. Specifically,

cot(η0) =
RP (0)[RQ(π)X ′

P (0) − rP (0)X ′

Q(0)]

rP (0)[RQ(π)RP (0) + rP (0)RQ(0)]
− X ′

P (0)

rP (0)
and

cot(ω0) =
RQ(π)X ′

P (0) − rP (0)X ′

Q(0)

RQ(π)RP (0) + rP (0)RQ(0)
,

where η0 is the angle of inclination of the tangent line at the basepoint nearer to P and ω0

is that of the farther basepoint’s tangent.

Proof. Our first goal in this proof is to find relationships among the functions rP , RP , and
RQ and their derivatives. Using the identities of the X-ray functions,

XP (ϕ) = RP (ϕ) − rP (ϕ),(1)

XQ(ϕ) = RQ(ϕ) +RQ(ϕ+ π),(2)

we will write these relationships as a linear system of equations and then solve this system.
Let us begin by placing our point P at the origin and letting Q lie along the x-axis to

the right of P ; thus, ℓ becomes the x-axis. Per our convention, we shall use ϕ to denote the
measure of the angle between the positive x-axis and a ray from P , and we shall denote by
ψ the measure of the angle between the positive x-axis and a ray from P (thus, π−ψ is the
measure of the angle from this ray to the negative x-axis). Finally, we let |PQ| represent
the distance from P to Q along the x-axis.

Consider a point γ on the nearside of K from P (see Figure 6). If we consider the triangle
△γPQ, the Law of Sines gives us the following relation:

(3)
RQ(π − ψ)

sin(ϕ)
=

rP (ϕ)

sin(π − ψ)

(
=
rP (ϕ)

sin(ψ)

)
.

After cross-multiplication and differentiation with respect to ψ, we find that

ϕ′(ψ)[rP (ϕ) cos(ϕ) + r′P (ϕ) sin(ϕ)] = RQ(π − ψ) cos(ψ) − R′

Q(π − ψ) sin(ψ),

Figure 6. γ is on the nearside of ∂K with respect to P .

or, after solving for ϕ′(ψ),

ϕ′(ψ) =
RQ(π − ψ) cos(ψ) −R′

Q(π − ψ) sin(ψ)

rP (ϕ) cos(ϕ) + r′P (ϕ) sin(ϕ)
.

Now, since these functions are continuous, we may evaluate them at ψ = π. Noticing from
Figure 6 that ϕ = 0 when ψ = π, we have

(4) ϕ′(π) = −RQ(0)

rP (0)
.

Looking again at Figure 6, we may also use the Law of Cosines to obtain

RQ(π − ψ)2 = |PQ|2 + rP (ϕ)2 − 2|PQ|rP (ϕ) cos(ϕ),

which, after differentiation with respect to ψ, becomes

−2RQ(π−ψ)R′

Q(π−ψ) = 2rP (ϕ)r′P (ϕ)ϕ′(ψ)−2|PQ|[r′P (ϕ) cos(ϕ)ϕ′(ψ)−rP (ϕ) sin(ϕ)ϕ′(ψ)],

or

R′

Q(π − ψ) = −ϕ′(ψ)
rP (ϕ)r′P (ϕ) − |PQ|[r′P (ϕ) cos(ϕ) − rP (ϕ) sin(ϕ)]

RQ(π − ψ)
.

Evaluating at ψ = π and substituting from (4), we find

R′

Q(0) = [rP (0) − |PQ|]r
′

P (0)

rP (0)
,

or, noting that rP (0) +RQ(π) = |PQ|,

(5) R′

Q(0) = −RQ(π)
r′P (0)

rP (0)
.

Now, suppose γ is on the farside of K from P (see Figure 7). Using the Law of Sines and
a similar calculation to that used to reach (4), remembering that ψ is now 0 when ϕ = 0,

we find that

(6) ψ′(0) =
RP (0)

RQ(π)
.

Figure 7. γ is on the farside of ∂K with respect to P .

Using the Law of Cosines (with RQ(π − ψ) expressed in terms of ϕ), differentiating with
respect to ϕ, evaluating at ϕ = 0, and substituting RQ(0) for RP (0) − |PQ|, we get

(7) R′

Q(π) = −RQ(0)
R′

P (0)

RP (0)
.

We can rewrite (5) and (7) as:

rP (0)R′

Q(0) = −RQ(π)r′P (0) and(8)

RP (0)R′

Q(π) = −RQ(0)R′

P (0),(9)

from which we can form the following linear system of equations by adding RQ(π)R′

P (0) to
each side of (8) and RP (0)R′

Q(0) to each side of (9) and then using identities (1) and (2):

RQ(π)R′

P (0) + rP (0)R′

Q(0) = RQ(π)[R′

P (0) − r′P (0)] = RQ(π)X ′

P (0)

− RQ(0)R′

P (0) +RP (0)R′

Q(0) = RP (0)[R′

Q(0) +R′

Q(π)] = RP (0)X ′

Q(0).

We translate this system into a matrix equation:
[

RQ(π) rP (0)
−RQ(0) RP (0)

] [
R′

P (0)
R′

Q(0)

]
=

[
RQ(π)X ′

P (0)
RP (0)X ′

Q(0)

]
.

Notice that det

(
RQ(π) rP (0)
−RQ(0) RP (0)

)
= RQ(π)RP (0) + rP (0)RQ(0) > 0 since rP (0), RP (0),

RQ(0), and RQ(π) are positive distances. Thus, we may invert our matrix to solve the

equation, which yields:

R′

P (0) =
RP (0)[RQ(π)X ′

P (0) − rP (0)X ′

Q(0)]

RQ(π)RP (0) + rP (0)RQ(0)
and

R′

Q(0) =
RQ(π)[RQ(0)X ′

P (0) +RP (0)X ′

Q(0)]

RQ(π)RP (0) + rP (0)RQ(0)
.

Using identities (1) and (2) again, we also have

r′P (0) =
RP (0)[RQ(π)X ′

P (0) − rP (0)X ′

Q(0)]

RQ(π)RP (0) + rP (0)RQ(0)
−X ′

P (0) and

R′

Q(π) = X ′

Q(0) −
RQ(π)[RQ(0)X ′

P (0) +RP (0)X ′

Q(0)]

RQ(π)RP (0) + rP (0)RQ(0)
.

The following equations are taken from page 12 of [8] and are ultimately derived from
page 85 of [1]:

r′P (ϕ) = rP (ϕ) cot(ηϕ − ϕ) and

R′

P (ϕ) = RP (ϕ) cot(ωϕ − ϕ),

where ηϕ is the angle of inclination of the line tangent to ∂K at (rP (ϕ), ϕ) and ωϕ is the
angle of inclination of the tangent line at (ϕ,RP (ϕ)). Using these equations, it is a simple
matter to solve for the angles of inclination of the tangent lines at the baseline; that is, when
ϕ = 0. The angles can also be similarly calculated using an analogous equation with Q as
the source point:

R′

Q(ψ) = RQ(ψ) cot(νψ − ψ),

where νψ is the angle of inclination of the tangent line at (ψ,RQ(ψ)), and taking ψ = π for
the basepoint nearer to P and ψ = 0 for the farther basepoint from P . �

Using the derived formulae on our example from above, we find that the tangent lines at
the basepoints of our shadow body are in fact vertical.

4. Computing Curvature at Basepoints

We continue by computing the curvature of the body at the basepoints. The procedure
for this computation is borrowed greatly from §6 of [8]. Computing the curvature will not
directly impact the process we shall later design to construct an approximation of a shadow
body; however, it will allow us to immediately check whether or not our shadow body will
have the desired sign of curvature at its basepoints.

Theorem. Let K be a convex body (C2 near its basepoints), and let P /∈ K, Q ∈ int K such
that P and Q lie on the x-axis with P to the left of Q. Then the curvature of ∂K at each of
K’s basepoints is computable when RP (0)RQ(π) 6= rP (0)RQ(0).

Proof. From page 91 of [1] we have the following identity:

(10) K(f + g) =
f + g

f
Kf +

f + g

g
Kg − 2fg

(
f ′

f
− g′

g

)2

We can apply this identity to (1) and (2) to obtain

KXP = K(RP − rP ) =
XP

RP

KRP − XP

rP
KrP + 2RP rP

(
R′

P

RP

− r′P
rP

)2

and(11)

KXQ = K(RQ + R̃Q) =
XQ

RQ

KRQ +
XQ

R̃Q

KR̃Q − 2RQR̃Q

(
R′

Q

RQ

−
R̃′

Q

R̃Q

)2

,(12)

where R̃Q(ϕ) = RQ(ϕ+π). If we evaluate these equations at 0, we see that this is a system of

two equations in four unknowns, namely, KRP (0), KrP (0), KRQ(0), and KR̃Q(0) = KRQ(π).
Now, since curvature is independent of parametrization, the curvature of ∂K at a basepoint

will be the same whether measured from P or from Q. Taking into account changes in
concavity, this fact gives us

KRP (0)

(RP (0)2 +R′

P (0)2)
3

2

= κRP (0) = κRQ(0) =
KRQ(0)

(RQ(0)2 +R′

Q(0)2)
3

2

and

KrP (0)

(rP (0)2 + r′P (0)2)
3

2

= κrP (0) = −κRQ(π) = − KRQ(π)

(RP (π)2 +R′

P (π)2)
3

2

.

Having evaluated (11) and (12) at 0, we can now make substitutions for KRQ(0) and KRQ(π)
to get

KXP (0) =
XP (0)

RP (0)
KRP (0) − XP (0)

rP (0)
KrP (0) + 2RP (0)rP (0)

(
R′

P (0)

RP (0)
− r′P (0)

rP (0)

)2

and

KXQ(0) =
XQ(0)

RQ(0)
KRP (0) ·

(
RQ(0)2 +R′

Q(0)2

RP (0)2 +R′

P (0)2

) 3

2

− XQ(0)

RQ(π)
KrP (0) ·

(
RQ(π)2 +R′

Q(π)2

rP (0)2 + r′P (0)2

) 3

2

− 2RQ(0)RQ(π)

(
R′

Q(0)

RQ(0)
−
R′

Q(π)

RQ(π)

)2

.

However, we can simplify this system using the fact that

(
RQ(0)2 +R′

Q(0)2

RP (0)2 +R′

P (0)2

) 3

2

=

(
RQ(0)

RP (0)

)3




1 +
(
R′

Q
(0)

RQ(0)

)2

1 +
(
R′

P
(0)

RP (0)

)2




3

2

=

(
RQ(0)

RP (0)

)3(
1 + cot(ν0 − 0)2

1 + cot(ω0 − 0)2

)
(from the equations of [1])

=

(
RQ(0)

RP (0)

)3

· 1 (since ν0 = ω0).

We can similarly compute that
(
RQ(π)2+R′

Q(π)2

rP (0)2+r′
P

(0)2

) 3

2

=
(
RQ(π)

rP (0)

)3

.

Using these simplifications, we translate the system into matrix equation form:

[
XP (0)
RP (0)

−XP (0)
rP (0)

XQ(0)RQ(0)2

RP (0)3
−XQ(0)RQ(π)2

rP (0)3

][
KRP (0)
KrP (0)

]
=


 KXP (0) − 2RP (0)rP (0)

(
R′

P
(0)

RP (0)
− r′

P
(0)

rP (0)

)2

KXQ(0) + 2RQ(0)RQ(π)
(
R′

Q
(0)

RQ(0)
− R′

Q
(π)

RQ(π)

)2


 .

We can now solve the system for KRP (0) and KrP (0) using matrix inversion, assuming
that

0 6= det

(
XP (0)
RP (0)

−XP (0)
rP (0)

XQ(0)RQ(0)2

RP (0)3
−XQ(0)RQ(π)2

rP (0)3

)

=
XP (0)XQ(0)RQ(0)2

RP (0)rP (0)3
− XP (0)XQ(0)RQ(π)2

RP (0)3rP (0)

=
XP (0)XQ(0)

RP (0)rP (0)

((
RQ(0)

RP (0)

)2

−
(
RQ(π)

rP (0)

)2
)
.

That is, when
(
RQ(0)

RP (0)

)2

6=
(
RQ(π)

rP (0)

)2

, or

RP (0)RQ(π) 6= rP (0)RQ(0).

Note that we did not solve for curvature itself but rather for the values of the curvature
operator at the basepoints of the convex body. However, it is a simple matter to find the
signed curvature from these values using the formulae from our definitions section. �

Using the curvature formula just derived to test our example body from above, we see
that the signed curvature at the nearside basepoint of the shadow body (as measured from
P ) is approximately −0.319666 and the signed curvature at the farside basepoint is about
0.321355. This confirms that our shadow body is indeed convex at its basepoints.

5. The Stable Manifold Theorem

Before attempting to create an algorithm which constructs a shadow body, we should
consider the mathematical principles that lead us to believe such a construction is possible.
Let us begin with the Stable Manifold Theorem:

Theorem. Let V be an open subset of R
2, and let f : V → R

2 be a Ck function with the

fixed point p ∈ V . Suppose that the eigenvalues of Df(p) =

[
∂fx

∂x
(p) ∂fx

∂y
(p)

∂fy

∂x
(p) ∂fy

∂y
(p)

]
are λ and µ,

where |λ| < 1 < |µ|.
Then there exist Ck curves WS and WU , respectively tangent to the eigenspaces Eλ and Eµ

of Df(p) at p, and an open subset V ′ ⊆ V such that for every x ∈ WS∩V ′, limj→∞ f j(x) = p
and for every x ∈ WU ∩ V ′, limj→∞ f−j(x) = p, where f j(x) is the j-fold application of the
function f (and f−j that of its inverse) to the point x.

We call WS the stable manifold for f at p, and we call WU the corresponding unstable
manifold. Notice that if a curve W is the stable manifold for f at p, then it is also the
unstable manifold for f−1 at p.

If K is a convex body with P /∈ K and Q ∈ intK, and P is at the origin, consider the
following functions defined for vectors v = (vx, vy) ∈ R

2 − {0}:

T+
P (v) = v

(
1 +

XP (arctan vy

vx
)

||v||

)

T−

P (v) = v

(
1 −

XP (arctan vy

vx
)

||v||

)
.

Purposely confusing vectors and points for the moment, we notice that each function simply
moves a point a distance equal to XP (ϕ) along the line connecting the point with P , where
ϕ is the angle of inclination of that line. We may also remark that T−

P = (T+
P )−1 for points

farther than XP (ϕ) from P .
Now, translate the plane so that Q lies at the origin, and consider the following map

defined for vectors v ∈ R
2 − {0}:

TQ(v) = v

(
1 −

XQ(arctan vy

vx
)

||v||

)
.

This map takes a point and moves it toward (and possibly beyond) Q a distance equal to
XQ(ψ) along the line connecting the point with Q, where ψ is the angle of inclination of
that line. TQ also acts as its own inverse for points within XQ(ψ) of Q. Indeed, for our
purposes we need only consider the intersection of {v ∈ R

2 | dist(P, v) > XP (arctan vy

vx
)}

and {v ∈ R
2 | 0 < dist(Q, v) < XQ(arctan vy

vx
)} as the domain for each of these three maps.

Now, consider a point p on the farside boundary of our example body: apply to p the
function f(p) = (TQT

−

P )2(p). The point will move towards the basepoint under this map
(see Figure 8). Applying f again will move the point closer to the basepoint. Now, since

Figure 8. Application of f = (TQT
−

P )2 to a point on the farside of the body.

the basepoint is itself fixed under this map, we have behavior matching the conclusion of
the Stable Manifold Theorem. We now predict f−1(p) = (T+

P TQ)2(p) will exhibit unstable
manifold-like behavior; indeed, points on the farside of the body tend away from the farside
basepoint (see Figure 9). Moreover, the vertices of our body act as fixed points under these
maps, and points on the farside of the body tend away from and towards the vertices under
the maps f and f−1, respectively. Thus, repeated application of f−1 to points near the
farside basepoint of our body will allow us to generate a good reconstruction of the original
body’s farside.

Figure 9. Application of f−1 = (T+
P TQ)2 to a point on the farside of the body.

Now, suppose there exists a shadow body. It should also have similar behavior under the
maps f and f−1. Indeed, as our algorithm will show us, it exhibits the opposite behavior at
its farside basepoint: points tend to move toward the basepoint under f−1 and away from
it under f (this is accounted for by the lack of horizontal symmetry of the shadow body—
after constructing the shadow body’s approximation, the reader might find it an interesting
exercise to apply f and f−1 to various points on the approximate body and to observe their
behavior under these maps). We could exploit this behavior by applying the map f to points
on the shadow body near its farside basepoint, but we do not actually know any points on the
shadow body except its basepoint. Therefore, we use a local approximation of the shadow
body near the basepoint—the farside tangent line—and apply the function to points on the
tangent line within a small distance of the basepoint.

6. Construction of an Approximate Shadow Body

We now attempt to construct an approximation of a shadow body (which we shall call S)
along the lines of [8] and [9]. In doing so, we will utilize the example convex body introduced
previously. First, we shall create points along the tangent lines of ∂S at each basepoint of
S. The points are placed at the intersection of the tangent lines and rays emanating from P
at uniformly distributed angles between 0 and ε, an angle chosen beforehand as a program
parameter. We operate on each point on every step of the algorithm.

The basic idea of the construction can be briefly described as “chord-chasing”; alternating
between the sources P and Q, we move each point across the expected shadow body along the

line connecting it and the source in question, using the function f(p) = (TQT
−

P )2(p) (actually,
the algorithm whose code is given below only applies the function f once per iteration, but
the end result will be the same). As we iterate this process, our points will start moving
away from the basepoints and towards the vertices of the shadow body (as the shadow body’s
farside boundary turns out to be the unstable manifold of f at the farside basepoint). The
iterated points should move along the boundary of the expected shadow body, allowing us
to make a rough sketch of the body’s shape. We terminate the algorithm when a point
moves to or beyond the shadow body’s vertex (failure to terminate the algorithm generates
interesting results—namely, a reconstruction of the original body—but does not help our
attempt to approximate the shadow body). Finally, we use Mathematica’s Interpolation

function to create a twice-differentiable curve that matches the final locations of the points.
We then compare this interpolated body with our original body in the hopes of finding only
very small discrepancies with our original X-ray data.

In order to generate a similar reconstruction for the nearside, we apply a similar map to
points on the tangent line at the nearside basepoint; however, we instead opt to record the
locations of the farside points after each application of T−

P ; this will also construct a nice
approximation of the shadow body’s nearside.

Mathematica code for the algorithm as well as post-algorithmic plots and checks of data
can be found in the appendix.

Figure 10. The result of running the reconstruction algorithm for 37 itera-
tions with an initial angle of .001 and with n = 1000.

A numerical comparison of the X-rays yielded the following results, though with messages
indicating the failure of Mathematica’s NIntegrate function to converge to prescribed accu-
racy near some points (here, X̂P and X̂Q are the X-ray functions of the interpolated shadow

Figure 11. The interpolation of the constructed shadow body, in blue, plot-
ted against the original body, in red.

Figure 12. The X-ray functions from P of the original body, in red, and the
interpolated shadow body, in blue.

body from P and Q, respectively):

∫ arccos
“

2
√

5

21

”

− arccos
“

2
√

5

21

”

|XP (ϕ) − X̂P (ϕ)|dϕ ≈ 3.09774 · 10−6

∫ π
2

−
π
2

|XQ(ψ) − X̂Q(ψ)|dψ ≈ 0.000012992

It is clear from these error estimates that the X-ray bodies are quite similar, though this is
far from a proof that there is in fact a shadow body.

We have also plotted the curvature of the near and far sides of the interpolated shadow
body. These curvatures were computed by Mathematica using the formula given in the
definition of signed curvature. Notice that the plots behave nicely except near angles along
which lie the vertices of the body. However, this misbehavior is likely to be an artefact
of Mathematica’s Interpolation function as well as numerical error that comes from the
approximation of the basepoints and the use of points on a tangent line for our construction
instead of points on the shadow body itself.

Figure 13. The X-ray functions from Q of the original body, in red, and the
interpolated shadow body, in blue.

Figure 14. The curvatures of the far and near sides of the interpolated
shadow body, respectively in red and blue, for ϕ ∈ [−π

2
, π

2
].

7. Conclusion

Though we have not produced a rigorous proof of the existence of a shadow body for
our example body, we have gathered evidence for the plausibility of such a shadow body’s
existence. There are a number of directions open to us for future research.

We have only thus far considered a “bottom-up” construction of a shadow body—that
is, from the basepoints to the vertices. Ross and Tuite [8] also produced an algorithm for
the construction of a shadow body using a “top-down” procedure from the vertices to the
basepoints. It is nearly certain that such a method could be devised in the case when one of
the point sources is in the interior of the convex body. It would first require the computation
of tangent lines and curvature in neighborhoods of the vertices, computations we attempted
to produce but did not because of time restraints. Further, one might be able to “link”
the constructions generated by the two algorithms in order to avoid the misbehavior of the
curvature plot at the vertices (and which would likely be present at the basepoints in a
top-down construction). One may even be able to find a numerical convergence to zero of
the error between the original X-ray bodies and the X-rays of the constructed shadow body.

Another possible approach would be to find better approximations of the shadow body at
its basepoints. For our research we only considered tangent lines of the shadow body, but
Siefken and Spargo were able to compute the second derivatives of the shadow body at its
basepoints [9], which would allow us to approximate the shadow body with more accurate
Taylor polynomial curves. However, there are some obstacles in this approach which must
be overcome. The higher-order derivatives quickly become very difficult to compute, though
it should be theoretically possible to compute as many as one might desire. Still, we would
need the Taylor series in order to find points precisely on the shadow body instead of points
very near it; this is problematic in that it is unknown whether such a Taylor series would
have a nonzero radius of convergence. Finally, even if an analytic curve is found which
is a piece of a shadow body, the algorithm described above as well as those designed and
implemented in [8] and [9] may ultimately create an object which is not a compact planar
region, let alone a convex region that may be properly called a shadow body.

The field of X-ray tomography has lain open for more than 25 years; the time we had
to perform our research lasted barely more than eight weeks. Far from making significant
progress in finding a solution to our problem, we have just begun to scratch the surface of
the subtleties that abound in this topic of study. We hope to continue our research on this
problem and eventually to come to a definitive solution to the question of the uniqueness of
a convex body given its point X-rays from two sources.

Appendix A. Code for the Construction Algorithm

The following is Mathematica code that may be used to conduct the algorithm for con-
structing a shadow body of the example lens from above. Note that this code may also be
used to reconstruct the original body as well as to construct other bodies, though it is best
suited for the example used in this paper. Note that this code was written on Mathematica
6.0; it will not work unmodified on earlier versions. The functions that will need modification
are likely to be the plotting functions.

(* Algorithm for the Construction of a Shadow Body from a Lens-shaped

Convex Body --- by Chris Pryby --- written for Mathematica 6.0 *)

Q = 10; (* x-coordinate of second point *)

(* lens -- intersection of two circles -- only guaranteed to work when Q is

is centered between the vertices of the lens *)

radius1 = 3;

center1 = 8;

radius2 = 3;

center2 = 12;

init = .001; (* initial angle of approximation - the smaller, the better,

but you’ll need more iterations *)

n = 250; (* there will be 2n+1 points plotted initially *)

iterations = 65; (* the more iterations, the more "reconstructed" the body

will become, but too many can lead to a collapse to the

stable manifolds near the basepoints and vertices --

this is accounted for since the algorithm stops if any

point goes past the vertex *)

doshadow = 1; (* 0 for the original body, 1 for the shadow body *)

showall = 1; (* 0 for only final plot, 1 for plots of all iterations *)

(* ---------- only modify the above values! ---------- *)

lensrightP[t_] = center1*Cos[t] + Sqrt[center1^2*Cos[t]^2 - (center1^2 -

radius1^2)];

dlensrightP[t_] = D[lensrightP[t], t];

lensleftP[t_] = center2*Cos[t] - Sqrt[center2^2*Cos[t]^2 - (center2^2 -

radius2^2)];

dlensleftP[t_] = D[lensleftP[t], t];

lensrightQ[t_] = (center1 - Q)*Cos[t] + Sqrt[(center1 - Q)^2*Cos[t]^2 -

((center1 - Q)^2 - radius1^2)];

dlensrightQ[t_] = D[lensrightQ[t], t];

lensleftQ[t_] = (center2 - Q)*Cos[t] - Sqrt[(center2 - Q)^2*Cos[t]^2 -

((center2 - Q)^2 - radius2^2)];

dlensleftQ[t_] = D[lensleftQ[t], t]*Sign[lensleftQ[t]];

(* finds angles on which the vertices lie by solving for the intersection of

the two circles *)

Print["This is the angle of inclination of the support line from P:"]

supportangle = t /. Quiet[Solve[lensrightP[t] == lensleftP[t], t]][[2]]

(* plots the original body *)

Print["This is the original body with P centered at the origin:"]

PolarPlot[{lensleftP[t], lensrightP[t]}, {t, -supportangle, supportangle}]

(* X-ray functions for lens *)

F[t_] = If[Mod[t,\[Pi]] >= \[Pi]/2, Mod[t,\[Pi]] - \[Pi], Mod[t,\[Pi]]];

lensXP[t_] = lensrightP[t] - lensleftP[t];

dlensXP[t_] = D[lensXP[t], t];

ddlensXP[t_] = D[dlensXP[t], t];

lensXQ[t_] = lensrightQ[t] - lensleftQ[t];

dlensXQ[t_] = D[lensXQ[t], t];

ddlensXQ[t_] = D[dlensXQ[t], t];

Print["This is the distance from P to each vertex:"]

supportradius = lensrightP[supportangle]

(* plots the X-ray bodies, from P and Q, of the original body *)

Print["This is the X-ray function of the original body from P:"]

PolarPlot[lensXP[F[t]], {t, -supportangle, supportangle}]

Print["This is the X-ray function of the original body from Q:"]

PolarPlot[lensXQ[F[t]], {t, -\[Pi]/2, \[Pi]/2}]

(* Falconer’s lemma - used for finding basepoints of potential shadow

body *)

p1 = lensrightP[0];

p2 = lensrightQ[0];

q1 = lensleftP[0];

q2 = lensleftQ[0];

m = p1 - q1;

A = Q;

B = p1*Log[Abs[p1]] - p2*Log[Abs[p2]] - q1*Log[Abs[q1]] + q2*Log[Abs[q2]];

f[t_] = t*Log[Abs[t]] - (t - m)*Log[Abs[t - m]];

g[t_] = f[t] - f[t - A];

Print["This is the plot of the Falconer g-function against y = B:"]

Plot[{g[t], B}, {t, -Q, 2*Q}, PlotStyle -> {Directive[Red, Thick],

Directive[Black, Thick, Dashed]}]

(* the result of the FindRoot operation will give the approximate right

basepoint for the potential shadow body *)

Print["These are the possible distances for p_1 given by Falconer’s Lemma:"]

rightbasepoint1 = t /. FindRoot[G[t] == B, {t, .999*p1}][[1]]

rightbasepoint2 = t /. FindRoot[G[t] == B, {t, 1.25*p1}][[1]]

(* This function of a point will shift the origin of polar coordinates

measurements for the point; e.g., if P has polar coordinates (t,r) when

measuring from the cartesian point O=(0,0), and we want to see P’s polar

coordinates when measuring from the cartesian point Q=(2,-5), we would

use ShiftOrigin[t,r,2,-5]. If the point would end up in the second or

third quadrants, we add \[Pi] to its resultant angle to account for the

fact that ArcTan’s range is (-\[Pi]/2,\[Pi]/2). *)

ShiftOrigin[t_, r_, h_, k_] = {If[r*Cos[t] - h < 0, \[Pi], 0] +

ArcTan[(r*Sin[t] - k)/(r*Cos[t] - h)],

Sqrt[(r*Cos[t] - h)^2 + (r*Sin[t] - k)^2]};

(* since the tangent line is vertical, we start by finding the intersection

of the line x == 9 (which is 9/cos(t)) with angles i*init/n between 0 and

initial angle init, i from -n to n *)

XP[t_] = lensXP[F[t]];

XQ[t_] = lensXQ[F[t]];

xshift = Q;

yshift = 0;

rightbasepoint = If[doshadow == 0, rightbasepoint1, rightbasepoint2];

leftbasepoint = rightbasepoint - XP[0];

(* sets up initial set of points to manipulate *)

(* the reconstruction collapses if you remove the nearside points and are

reconstructing the original body; it also collapses if you remove the

farside points and are constructing the shadow body *)

PointSet = {};

(* initial points on nearside tangent line for original body, farside for

shadow body *)

If[doshadow == 0,

For[i = -n, i <= n, i++,

PointSet = Union[PointSet, {{i*init/n, (rightbasepoint -

XP[0])Cos[i*init/n]}}]],

For[i = -n, i <= n, i++,

PointSet = Union[PointSet, {{i*init/n,

rightbasepoint*Cos[i*init/n]}}]]

]

(* initial plot of tangent lines *)

If[showall != 0, Print["This is the initial setup of the points to be

manipulated -- they lie on a tangent line of the

body:"], 0;]

If[showall != 0, ListPolarPlot[PointSet, PlotRange -> All, PlotStyle ->

Directive[Thick, Black]], 0;]

If[showall != 0, Print["These are plots of successive iterations of the

algorithm -- every two plots demonstrates a

complete iteration:"], 0;]

stop = 0;

(* chord chasing procedure *)

For[i = 1, i <= iterations && stop == 0, i++,

(* prints the iteration number *)

If[showall == 1, Print[i], 0];

(* forces the computer to store points as numerical approximations

instead of exact expressions -- speeds up the process *)

PointSet = N[PointSet];

(* adds (or subtracts) the X-ray data obtained from P -- subtract X_P if

curvature from P is positive *)

For[j = 1, j <= Length[PointSet], j++,

angle = PointSet[[j]][[1]];

radius = PointSet[[j]][[2]];

(* determining whether to add or subtract XP is currently

hardcoded for the lens -- it should be easy if we have a shape

with two vertices which the support lines from P intersect, but

otherwise it gets difficult (for the shadow body, that is) *)

PointSet[[j]] = {angle, radius + If[radius > supportradius, -1, 1]*

XP[angle]};

stop = If[doshadow == 0, If[radius >= supportradius, 1, 0],

If[radius <= supportradius, 1, 0]]

];

If[showall == 1, Print[ListPolarPlot[PointSet, PlotRange -> All,

PlotStyle -> Directive[Thick, Black]]], 0;];

(* stores PointSet after adding/subtracting from P for future use in

superimposing reconstructed nearside & farside *)

PrevPointSet = PointSet;

(* shifts origin to Q *)

For[j = 1, j <= Length[PointSet], j++,

angle = PointSet[[j]][[1]];

radius = PointSet[[j]][[2]];

PointSet[[j]] = ShiftOrigin[angle, radius, xshift, yshift]

];

(* subtracts X-ray data obtained from Q -- always should subtract since

the points must move inwards towards Q *)

For[j = 1, j <= Length[PointSet], j++,

angle = PointSet[[j]][[1]];

radius = PointSet[[j]][[2]];

PointSet[[j]] = {angle, radius - XQ[angle]}

];

(* shifts origin back to P *)

For[j = 1, j <= Length[PointSet], j++,

angle = PointSet[[j]][[1]];

radius = PointSet[[j]][[2]];

PointSet[[j]] = ShiftOrigin[angle, radius, -xshift, -yshift]

];

If[showall == 1, Print[ListPolarPlot[PointSet, PlotRange -> All,

PlotStyle -> Directive[Thick, Black]]], 0;];

(* chord-chasing terminates if point passes vertex - should rerun the

algorithm with one less iteration to get a more accurate

interpolation *)

If[stop != 0, Print["Process stopped because the reconstruction crossed

a vertex on iteration ", i], 0;]

](* close for-loop *)

ListPolarPlot[{PointSet, PrevPointSet}, PlotRange -> All, PlotStyle ->

Directive[Thick, Black]]

(* interpolation of data from P *)

shadowrightP = Interpolation[PointSet, InterpolationOrder -> 2];

dshadowrightP[t_] = D[shadowrightP[t], t];

shadowleftP = Interpolation[PrevPointSet, InterpolationOrder -> 2];

dshadowleftP[t_] = D[shadowleftP[t], t];

shadowXP[t_] = shadowrightP[t] - shadowleftP[t];

dshadowXP[t_] = D[shadowXP[t], t];

Print["This is the interpolated shadow body plotted against the original

body."]

Quiet[PolarPlot[{lensrightP[t], lensleftP[t], shadowrightP[t],

shadowleftP[t]}, {t, -supportangle, supportangle},

PlotStyle -> {Directive[Red, Thick], Directive[Red, Thick],

Directive[Blue, Thick, Dashed], Directive[Blue, Thick, Dashed]}]]

Print["This is the X-ray from P of the interpolated shadow body plotted

against the X-ray of the original body."]

Quiet[PolarPlot[{lensXP[t], shadowXP[t]}, {t, -supportangle,

supportangle}, PlotStyle -> {Directive[Red, Thick],

Directive[Blue, Thick, Dashed]}]]

(* shifting data to originate from Q *)

QSet = {};

PrevQSet = {};

For[i = 1, i <= Length[PointSet], i++,

angle = PointSet[[i]][[1]];

radius = PointSet[[i]][[2]];

QSet = Union[QSet, {ShiftOrigin[angle, radius, Q, 0]}]

]

For[i = 1, i <= Length[PrevPointSet], i++,

angle = PrevPointSet[[i]][[1]];

radius = PrevPointSet[[i]][[2]];

PrevQSet = Union[PrevQSet, {ShiftOrigin[angle, radius, Q, 0]}]

]

(* interpolation of data from Q -- domain of left side of shadow body from Q

was altered, and the leftside data was added instead of subtracted to get

X_Q, but the idea is what counts *)

shadowrightQ = Interpolation[QSet, InterpolationOrder -> 2];

dshadowrightQ[t_] = D[shadowrightQ[t], t];

tempshadowleftQ = Interpolation[PrevQSet, InterpolationOrder -> 2];

shadowleftQ[t_] = -tempshadowleftQ[t + \[Pi]];

dshadowleftQ[t_] = D[shadowleftQ[t], t]*Sign[shadowleftQ[t]];

shadowXQ[t_] = shadowrightQ[t] - shadowleftQ[t];

dshadowXQ[t_] = D[shadowXQ[t], t];

Print["This is the interpolated shadow body plotted against the original

body:"]

Quiet[PolarPlot[ {lensrightQ[t], lensleftQ[t], shadowrightQ[t],

shadowleftQ[t]}, {t, -\[Pi]/2, \[Pi]/2}, PlotStyle -> {Directive[Red,

Thick], Directive[Red, Thick], Directive[Blue, Thick, Dashed],

Directive[Blue, Thick, Dashed]}]]

Print["This is the X-ray from Q of the interpolated shadow body plotted

against the X-ray of the original body:"]

Quiet[PolarPlot[{lensXQ[t], shadowXQ[t]}, {t, -\[Pi]/2, \[Pi]/2}, PlotStyle

-> {Directive[Red, Thick], Directive[Blue, Thick, Dashed]}]]

ddshadowleftP[t_] = D[shadowleftP[t], {t, 2}];

ddshadowleftQ[t_] = D[shadowleftQ[t], {t, 2}]*Sign[shadowleftQ[t]];

ddshadowrightP[t_] = D[shadowrightP[t], {t, 2}];

ddshadowrightQ[t_] = D[shadowrightQ[t], {t, 2}];

(* testing the values of the curvatures *)

Print["These are the curvatures of the original body at its basepoints as

computed by my formula (should be +/- 1/3):"]

matrixA = {{lensXP[0]/lensrightP[0], -lensXP[0]/lensleftP[0]},

{lensXQ[0]/lensrightQ[0]*((lensrightQ[0]^2 + dlensrightQ[0]^2)/

(lensrightP[0]^2 + dlensrightP[0]^2))^(3/2), -lensXQ[0]/

Abs[lensleftQ[0]]*((Abs[lensleftQ[0]]^2 + dlensleftQ[0]^2)/(

lensleftP[0]^2 + dlensleftP[0]^2))^(3/2)}};

matrixB = {lensXP[0]^2 + 2*dlensXP[0]^2 - lensXP[0]*ddlensXP[0] -

2*lensrightP[0]*lensleftP[0]*(dlensrightP[0]/lensrightP[0] -

dlensleftP[0]/lensleftP[0])^2, lensXQ[0]^2 + 2 dlensXQ[0]^2 -

lensXQ[0] ddlensXQ[0] + 2*lensrightQ[0]*Abs[lensleftQ[0]]*

(dlensrightQ[0]/lensrightQ[0] - dlensleftQ[0]/

Abs[lensleftQ[0]])^2};

kops = Inverse[matrixA].matrixB;

Print["Farside curvature:"]

kops[[1]]/(lensrightP[0]^2 + dlensrightP[0])^(3/2)

Print["Nearside curvature:"]

kops[[2]]/(lensleftP[0]^2 + dlensleftP[0])^(3/2)

Print["These are the curvatures of the shadow body at its basepoints as

computed by my formula:"]

matrixA = {{lensXP[0]/shadowrightP[0], -lensXP[0]/shadowleftP[0]},

{lensXQ[0]/shadowrightQ[0]*((shadowrightQ[0]^2 +

dshadowrightQ[0]^2)/(shadowrightP[0]^2 + dshadowrightP[0]^2))^

(3/2), -lensXQ[0]/Abs[shadowleftQ[0]]*((Abs[shadowleftQ[0]]^2 +

dshadowleftQ[0]^2)/(shadowleftP[0]^2 + dshadowleftP[0]^2))^(3/2)}};

matrixB = {lensXP[0]^2 + 2*dlensXP[0]^2 - lensXP[0]*ddlensXP[0] -

2*shadowrightP[0]*shadowleftP[0]*(dshadowrightP[0]/

shadowrightP[0] - dshadowleftP[0]/shadowleftP[0])^2, lensXQ[0]^2 +

2*dlensXQ[0]^2 - lensXQ[0]*ddlensXQ[0] + 2*shadowrightQ[0]*

Abs[shadowleftQ[0]]*(dshadowrightQ[0]/shadowrightQ[0] -

dshadowleftQ[0]/Abs[shadowleftQ[0]])^2};

kops = Inverse[matrixA].matrixB;

Print["Farside curvature:"]

kops[[1]]/(shadowrightP[0]^2 + dshadowrightP[0])^(3/2)

Print["Nearside curvature:"]

kops[[2]]/(shadowleftP[0]^2 + dshadowleftP[0])^(3/2)

Print["These are the approximate curvatures of the shadow body at the

basepoints as computed by Mathematica:"]

kshadowrightP[t_] = (shadowrightP[t]^2 + 2*dshadowrightP[t]^2 -

shadowrightP[t]*ddshadowrightP[t])/(shadowrightP[t]^2 +

dshadowrightP[t]^2)^(3/2);

kshadowleftP[t_] = (shadowleftP[t]^2 + 2*dshadowleftP[t]^2 - shadowleftP[t]*

ddshadowleftP[t])/(shadowleftP[t]^2 + dshadowleftP[t]^2)^

(3/2);

kshadowrightQ[t_] = (shadowrightQ[t]^2 + 2*dshadowrightQ[t]^2 -

shadowrightQ[t]*ddshadowrightQ[t])/(shadowrightQ[t]^2 +

dshadowrightQ[t]^2)^(3/2);

kshadowleftQ[t_] = (Abs[shadowleftQ[0]]^2 + 2*dshadowleftQ[t]^2 -

Abs[shadowleftQ[0]]*ddshadowleftQ[t])/

(Abs[shadowleftQ[0]]^2 + dshadowleftQ[t]^2)^(3/2);

Print["Farside curvature, parametrized by P:"]

kshadowrightP[0]

Print["Nearside curvature, parametrized by P:"]

kshadowleftP[0]

Print["Farside curvature, parametrized by Q:"]

kshadowrightQ[0]

Print["Nearside curvature, parametrized by Q:"]

kshadowleftQ[0]

Print["Here are plotted the approximate curvatures of the shadow body as

computed by Mathematica, parametrized by P:"]

Quiet[Plot[{kshadowrightP[t], kshadowleftP[t]}, {t, -supportangle,

supportangle}, PlotStyle -> {Directive[Thick, Red],

Directive[Thick, Blue, Dashed]}, PlotRange -> {-.35, .75}]]

Print["Here are some numerical computations of error between the original X-

rays and the X-rays of the interpolated shadow body:"]

NIntegrate[Abs[shadowXP[t] - lensXP[t]], {t, -ArcCos[2*Sqrt[5/21]],

ArcCos[2*Sqrt[5/21]]}]

NIntegrate[Abs[shadowXP[t] - lensXP[t]]/Sin[t], {t, -ArcCos[2*Sqrt[5/21]],

ArcCos[2 Sqrt[5/21]]}]

NIntegrate[Abs[shadowXQ[t] - lensXQ[t]], {t, -\[Pi]/2, \[Pi]/2}]

NIntegrate[Abs[shadowXQ[t] - lensXQ[t]]/Sin[t], {t, -\[Pi]/2, \[Pi]/2}]

